The problem is to compute the correct path to fly a commercial jet airliner from a given origin to a given destination so as to minimize fuel-cost/flight-time. The model must account for winds aloft, which at 40,000 feet can be significant, and it must account for the nonplanarity of the earth, since long-haul flights cover a significant fraction of the circumference of the earth.

Let \( \rho \) denote the radius of the earth and let \( \theta \) and \( \phi \) denote the two angles in a standard spherical coordinate system. That is, \( \theta \) is an angle measured (in radians) east from some meridian and \( \phi \) is the angle down from the north pole. In these coordinates, the distance metric along the surface of the earth, or more precisely 40,000 feet above it, is given by

\[
 ds^2 = \rho^2 (d\phi^2 + \sin^2(\phi)d\theta^2). 
\]

Longitude \( \lambda \) is the same as \( \theta \) except that it is measured in degrees:

\[
 \lambda = \frac{180\theta}{\pi}. 
\]

Latitude \( \mu \) being measured up from the equator is the angle complementary to \( \phi \). It is also measured in degrees:

\[
 \mu = 90 - \frac{180\phi}{\pi}. 
\]

In terms of latitude and longitude, the distance metric is

\[
 ds^2 = \rho^2 \left( d\mu^2 + \cos^2(\pi\mu/180)d\lambda^2 \right) \pi/180. 
\]

We assume that winds aloft are known everywhere (at least approximately). Let \( w_\lambda(\lambda, \mu) \) denote the longitudinal component of the winds aloft at location \( (\lambda, \mu) \) and let \( w_\mu(\lambda, \mu) \) denote the latitudinal component. These two functions are assumed to be known.

The problem is to find a trajectory \( (\lambda(t), \mu(t)), 0 \leq t \leq T \), connecting a given initial location \( (\lambda_0, \mu_0) \) to a given final location \( (\lambda_T, \mu_T) \). We wish to minimize the total time \( T \) subject to an upper bound \( V_{\text{max}} \) on the airspeed (not the groundspeed):

\[
 \rho^2 \left( (\dot{\mu} - w_\mu(\lambda, \mu))^2 + \cos^2(\pi\mu/180)(\dot{\lambda} - w_\lambda(\lambda, \mu))^2 \right) \pi/180 \leq V_{\text{max}}. 
\]

The AMPL model for the resulting nonlinear optimization problem is shown in Figure 1.
# Wind fcns are defined externally
# in amplfunc.dll.
function wLat;
function wLon;

param pi := 4*atan(1);  # 3.14159...
param N := 100;  # discretization
param wind;  # a wind intensity factor

param rho := 4000;  # radius of Earth in miles
param Vmax := 550;  # max air-speed in miles/hr

# start in NYC:
param lon0 := -73;
param lat0 := 40.5;

# end in Honolulu:
param lonT := -158;
param latT := 21.3;

var T >= 0, := 6;  # total time in hrs

var lon {j in 0..N} := (j/N)*lonT + (1-j/N)*lon0;
var lat {j in 0..N} := (j/N)*latT + (1-j/N)*lat0;

var lonDot {j in 1..N} := (lon[j]-lon[j-1])/(T/N);
var latDot {j in 1..N} := (lat[j]-lat[j-1])/(T/N);

var wLatAvg {j in 1..N} := wLat(lat[j],lon[j]) + wLat(lat[j-1],lon[j-1])/2;
var wLonAvg {j in 1..N} := wLon(lat[j],lon[j]) + wLon(lat[j-1],lon[j-1])/2;

minimize time: T;

s.t. lonInit: lon[0] = lon0;
s.t. latInit: lat[0] = lat0;
s.t. lonFinl: lon[N] = lonT;
s.t. latFinl: lat[N] = latT;

s.t. speedLimit {j in 1..N}:
    ( latDot[j] - wind*wLatAvg[j] )^2 +
    cos(pi*(lat[j]+lat[j-1])/360)^2 *
    ( lonDot[j] - wind*wLonAvg[j] )^2 <=
    (180/pi)^2 * (Vmax/rho)^2;

---

FIGURE 1. The AMPL model `flightpath5.mod`.  

---